Lab 4: Inverse taper edge coupler

Objective:

Design a Si waveguide to optimize single mode coupling to an optical fiber.

Background:

The silicon photonic edge coupler is an essential component of integrated photonic circuits, allowing efficient coupling of light between a photonic chip and external optical fibers. These couplers are designed to convert light propagating through the chip's waveguides to a free-space radiation mode that can be coupled to a fiber optic cable. By doing so, they enable high-speed data communication, optical sensing, and other applications in a compact and cost-effective manner. Silicon photonic edge couplers have been extensively studied and optimized over the years, resulting in various designs and fabrication techniques that offer high coupling efficiency, low loss, and broadband operation.



Procedure:

<u>Script the waveguide structure</u>

1. Initialize the script variables

First, we must initialize the Lumerical script variables. We will be changing three simulation factors: waveguide width, wavelength start, and wavelength stop. In the Lumerical model, create three variables as shown in figure 2. Additionally, initialize the simulation area with the *selectall* and *delete* commands.



Fig. 2: Initialized variables

2. Parameterize Si waveguide geometry

In this step, we must create a configurable geometry to create our Si waveguide. Assume 220 nm thick SOI with a symmetric 3 µm oxide cladding. Underneath the waveguide is a handle Si wafer. Set the Si width to be *waveguide_width*.

Fig 3. XY cross section of SOI waveguide. Z axis points out of the page.

```
waveguide_thickness = 0.22e-6;
waveguide_length = 3e-6;
cladding_thickness = 3e-6;
MFD = 10.4e-6;

addrect;
set('name', 'waveguide');
set('x', 0);
set('x span', waveguide_width);
set('y', 0);
set('y span', waveguide_thickness);
set('z min', -waveguide_length);
set('z max', 0);
set('material', 'Si (Silicon) - Palik');
set('override mesh order from material database', true);
set('mesh order', 1);

addrect;
set('name', 'cladding');
set('x', 0);
```

```
set('x span', 2*MFD+1e-6);
set('y', 0);
set('y span', waveguide_thickness + 2*cladding_thickness);
set('z min', -waveguide_length);
set('z max', 0);
set('material', 'SiO2 (Glass) - Palik');
set('alpha', 0.5);
set('override mesh order from material database', true);
set('mesh order', 2);

addrect;
set('name', 'handle_Si');
set('x', 0);
set('x span', 2*MFD+1e-6);
set('y max', -waveguide_thickness/2 - cladding_thickness);
set('y min',-waveguide_thickness/2 - cladding_thickness - 5e-6);
set('z min', -waveguide_length);
set('z max', 0);
set('material', 'Si (Silicon) - Palik');
set('override mesh order from material database', true);
set('mesh order', 1);
```

3. Define optical fiber index and geometry

Next, we create the optical fiber. We will use a common telecom fiber, SMF-28. The specs for core/cladding diameter can be found online, whereas the index can be approximated from the MFD.

```
n_core = 1.46813;
n_cladding = 1.45392;
fiber_gap = 1e-6;
core_diameter = 8.2e-6;
cladding_diameter = 125e-6;

addcircle;
set('name', 'fiber_core');
set('radius', core_diameter/2);
set('x', 0);
set('y', 0);
set('z min', fiber_gap);
set('z max', fiber_gap + 5e-6);
set('index', n_core);
set('alpha', 0.5);
set('override mesh order from material database', true);
set('mesh order', 3);

addcircle;
set('name', 'fiber_cladding');
set('radius', cladding_diameter/2);
set('x', 0);
set('y', 0);
```

```
set('z min', fiber_gap);
set('z max', fiber_gap + 5e-6);
set('index', n_cladding);
set('alpha', 0.25);
set('override mesh order from material database', true);
set('mesh order', 4);
```

4. Inject mode and create mode match monitors

Next, we create the waveguide mode and fiber mode monitors. We can input a mode source into the waveguide and place a power & mode expansion monitor in the fiber.

```
addmode;
set('name', 'mode');
set('injection axis', 'Z-Axis');
set('x', 0);
set('x span', 2*MFD);
set('y', 0);
set('y span', waveguide_thickness + 2*cladding_thickness);
set('z', -0.5e-6);
set('wavelength start', lambda_start);
set('wavelength stop', lambda_stop);
set('mode selection', 'fundamental TE mode');


addpower;
set('monitor type', '2D Z-Normal');
set('name', 'fiber_mode');
set('x', 0);
set('x span', 2*MFD);
set('y', 0);
set('y span', 2*MFD);
set('z', fiber_gap + 0.5e-6);

addmodeexpansion;
set('monitor type', '2D Z-Normal');
set('name', 'fiber_mode_expansion');
setexpansion('mode_overlap', 'fiber_mode');
set('x', 0);
set('x span', 2*MFD);
set('y', 0);
set('y span', 2*MFD);
set('z', fiber_gap + 1e-6);
set('mode selection', 'fundamental TE mode');
```

5. Define and optimize simulation area

Our penultimate task is to define the FDTD simulation area. We will use metal and symmetric boundary conditions where we can to minimize simulation time.

```
addfdtd;
```

```
set('dimension', '3D');
set('simulation time', 100e-15);
set('auto shutoff min', 1e-4);
set('x', 0);
set('x span', 2*MFD);
set('y', 0);
set('y span', 2*MFD);
set('z min', -1e-6);
set('z max', fiber_gap + 1e-6);
set('z min bc', 'metal');
set('y min bc', 'metal');
set('x min bc', 'anti-symmetric');
```

6. Declare and store result

Run the simulation to switch to analysis mode. Then, add a result for T under model -> Analysis



Fig 4. Result stored as *T*

Then, add a few lines of code to store the net transmission into our mode from the mode expansion monitor:

```
t = getresult('fiber_mode_expansion', 'expansion for mode_overlap');
T = t.T_net;
?T;
```

Optimize the result:

Waveguide Width Sweep:

1. Add new sweep in “Optimizations and Sweeps” tab
2. Add waveguide_width as parameter. Sweep from 0.05 to 0.35 µm.
3. Add T as a result and run the sweep

Benchmark the design:

Bandwidth Estimation

1. Enter the optimized result for waveguide_width
2. Change lambda_start to 1.31 µm and run the simulation
3. View the result

Follow-up questions:

1. Compare and contrast the performance of the edge coupler and the grating coupler. If you were a professional SiPh designer, what considerations would go into your selection process?
2. What is the expected ‘fabrication tolerance’ of the edge coupler? I.e. what are design factors that can change with the fabrication process, and how does that impact performance? What about with the grating?
3. We’ve only simulated the end of the edge coupler – later, we will simulate the taper structure, or how the waveguide gets to this design thickness. What design parameters go into creating a waveguide taper?