Lab 4: EME Directional Coupler Design

Objective:

- Understand the key design parameters of a directional coupler.
- Learn the basic setup and usage of the *Lumerical EME* solver.
- Design a 50/50 directional coupler splitter with *EME*. Evaluate fabrication tolerances and broadband behavior.

Background:

- Directional Coupler (DC)

Directional couplers are fundamental components in integrated photonic circuits, enabling controlled power transfer between adjacent waveguides through evanescent field coupling. When two waveguides are placed in close proximity over a certain interaction length, their optical modes overlap, allowing energy to transfer back and forth between them. By designing the coupling region appropriately, directional couplers can be tailored to function as power splitters, switches, or wavelength filters, playing a critical role in applications ranging from optical communication systems to quantum photonics.

- *Lumerical Eigenmode Expansion* (*EME*)

Lumerical EME is a software tool for computing the modes of waveguides and other structures. It expands the electric field distribution of a mode in terms of the eigenmodes of the waveguide or structure and computes the expansion coefficients using mode overlap integrals. EME can simulate the coupling between different eigenmodes and allows for the optimization of device designs.

Procedure:

<u>Simulation Setup</u>

1. Open MODE from the desktop shortcuts folder



2. Initialize simulation variables

   Create the following simulation variables. These will control the directional coupler geometry, simulation wavelength, and the EME spacing.

Name	Type	Explanation
`Lc`	Length	Length of the central straight waveguide sections for coupling
`Lw`	Length	Length of the four individual straight waveguide sections at the input and output ports
`Lb`	Length	X-span of the four individual bend sections
`wg_width`	Length	Width of the waveguide base
`wg_height`	Length	Height of the waveguide
`wg_angle`	Number	Sidewall angle of the waveguide
`gap`	Length	Gap between two coupling sections
`sep`	Length	Center-to-center distance between the two input ports
`wavelength`	Length	EME simulation wavelength
`n_cells`	Number	Number of EME cells in the coupling section
`n_cells_side`	Number	Number of EME cells in the port and bend sections on each side



3. Parameterize directional coupler geometry

   I used the following codes to define the Si waveguide structures of the directional coupler.

```
deleteall;

## Top Half
addstructuregroup;
set("name", "directional_coupler_top");

# Waveguide Top Left (Port 1)
addwaveguide;
set("name", "wg_top_left");
```

```
addtogroup("directional_coupler_top");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [-1e-6, sep/2; Lw, sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Bend Top Left
addwaveguide;
set("name", "bend_top_left");
addtogroup("directional_coupler_top");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw, sep/2; Lw+Lb/2, sep/2; Lw+Lb/2, gap/2+wg_width/2; Lw+Lb,
gap/2+wg_width/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Coupling Waveguide Top
addwaveguide;
set("name", "coupler_top");
addtogroup("directional_coupler_top");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+Lb, gap/2+wg_width/2; Lw+Lb+Lc, gap/2+wg_width/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Bend Top Right
addwaveguide;
set("name", "bend_top_right");
addtogroup("directional_coupler_top");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+Lb+Lc, gap/2+wg_width/2; Lw+(1+1/2)*Lb+Lc, gap/2+wg_width/2;
Lw+(1+1/2)*Lb+Lc, sep/2; Lw+2*Lb+Lc, sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Waveguide Top Right (Port 3)
addwaveguide;
set("name", "wg_top_right");
addtogroup("directional_coupler_top");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+2*Lb+Lc, sep/2; 2*Lw+2*Lb+Lc+1e-6, sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

## Bottom Half
addstructuregroup;
set("name", "directional_coupler_bottom");

# Waveguide Bottom Left (Port 2)
addwaveguide;
set("name", "wg_bottom_left");
```

```
addtogroup("directional_coupler_bottom");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [-1e-6, -sep/2; Lw, -sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Bend Bottom Left
addwaveguide;
set("name", "bend_bottom_left");
addtogroup("directional_coupler_bottom");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw, -sep/2; Lw+Lb/2, -sep/2; Lw+Lb/2, -gap/2-wg_width/2; Lw+Lb, -gap/2-
wg_width/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Coupling Waveguide bottom
addwaveguide;
set("name", "coupler_bottom");
addtogroup("directional_coupler_bottom");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+Lb, -gap/2-wg_width/2; Lw+Lb+Lc, -gap/2-wg_width/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Bend Bottom Right
addwaveguide;
set("name", "bend_bottom_right");
addtogroup("directional_coupler_bottom");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+Lb+Lc, -gap/2-wg_width/2; Lw+(1+1/2)*Lb+Lc, -gap/2-wg_width/2;
Lw+(1+1/2)*Lb+Lc, -sep/2; Lw+2*Lb+Lc, -sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");

# Waveguide Bottom Right (Port 4)
addwaveguide;
set("name", "wg_bottom_right");
addtogroup("directional_coupler_bottom");
set("base width", wg_width);
set("base height", wg_height);
set("base angle", wg_angle);
pole = [Lw+2*Lb+Lc, -sep/2; 2*Lw+2*Lb+Lc+1e-6, -sep/2];
set("poles", pole);
set("material", "Si (Silicon) - Palik");
```

Note that in Lumerical, the waveguide is defined by its cross-section (characterized by width, height, and sidewall angle) and its path (determined by poles). The path follows a Bézier curve, with the poles serving as its control points. For the bending section, the waveguide can be specified using the following geometry.



4. Define EME solver region

   Create an EME solver with three groups. Make the groups span across the input and output waveguide respectively. Space the FDE cells by approx. one wavelength. Include four ports on the four waveguide connecting to the directional coupler.

```
# EME Region
addeme;
set("solver type", "3D: X Prop");
set("background material", "SiO2 (Glass) - Palik");
set("x min", 0);
set("y", 0); set("y span", 8e-6);
set("z", 0); set("z span", 2e-6);
set("wavelength", wavelength);
set("number of modes for all cell groups", 20);
set("number of cell groups", 3);
set("cells", [n_cells_side; n_cells; n_cells_side]);
set("group spans", [Lw+Lb; Lc; Lw+Lb]);
set("subcell method",[1; 1; 1]);
set("display cells", true);

# EME Ports
select("EME::Ports::port_1");
set("port location", "left");
set("x", 0);
set("use full simulation span",0);
set("y min", -sep/2+wg_width/2+0.1e-6);
set("y max", sep/2+5e-6);
set("z", 0); set("z span", 5e-6);
set("mode selection", "fundamental TE mode");

select("EME::Ports::port_2");
set("port location", "left");
set("x", 0);
set("use full simulation span",0);
set("y max", sep/2-wg_width/2-0.1e-6);
set("y min", -sep/2-5e-6);
set("z", 0); set("z span", 5e-6);
set("mode selection", "fundamental TE mode");

addemeport;
select("EME::Ports::port_3");
set("port location", "right");
```

```
set("x", 2*Lw+2*Lb+Lc);
set("use full simulation span",0);
set("y min", -sep/2+wg_width/2+0.1e-6);
set("y max", sep/2+5e-6);
set("z", 0); set("z span", 5e-6);
set("mode selection", "fundamental TE mode");

addemeport;
select("EME::Ports::port_4");
set("port location", "right");
set("x", 2*Lw+2*Lb+Lc);
set("use full simulation span",0);
set("y max", sep/2-wg_width/2-0.1e-6);
set("y min", -sep/2-5e-6);
set("z", 0); set("z span", 5e-6);
set("mode selection", "fundamental TE mode");
```

Note that `set("subcell method",[1; 1; 1])` enables the CVCS (Continuously Varying Cross-Sectional Subcell) method for each cell (0 for none and 1 for CVCS).

Traditional EME approximates continuously varying structures, like tapers, with a staircase model, leading to non-physical reflections and inaccuracies. The usual fix — adding more expansion interfaces — increases computational time and memory usage. According to Lumerical, their CVCS method eliminates staircasing, offers good accuracy at a fraction of the time for longer structures.



Also note that the line `set("use full simulation span",0)` under each port ensures that the port does not span the entire yz-plane of the EME region. If you're using the GUI, this is equivalent to unchecking the "*use full simulation span*" option. In Lumerical EME, this option is checked by default, so if you're simulating a structure with multiple ports on one side, you must uncheck it. Otherwise, the port will occupy the entire side, preventing the separation of distinct ports on that side.



Another important point to note is that the position of each port is defined *using relative coordinates* by default. This means the coordinates are relative to the EME region. In our simulation, the EME region is located at y = 0 and z = 0, so the y and z coordinates of the ports directly match their actual spatial positions. However, if the EME region is not centered at the origin in the yz-plane, you need to account for the y and z offsets of the EME region when specifying port positions.



5. Other components:

Add a fine mesh.

```
# Fine Mesh
addmesh;
dr = 100e-9;
set("x min", 0);
set("x max", 2*Lw+2*Lb+Lc);
set("y", 0);
set("y span", 8e-6);
set("z", 0);
set("z span", 4e-6);
set("dx", dr);
set("dy", dr);
set("dz", dr);
```

Add a field monitor that spans the XY plane of the directional coupler to visualize the optical field distribution.

```
# Cross-Section Field Monitor
addemeprofile;
set("name", "profile");
set("x min", 0);
set("x max", 2*Lw+2*Lb+Lc);
set("y", 0);
set("y span", 8e-6);
set("z", 0);
```

You can download ***directional_coupler_eme.lms*** to check your setup.

EME Simulation

6. Insert starting conditions for the waveguide geometry

#	Name	Type	Value	Unit
1	Lc	Length	20	um
2	Lw	Length	5	um
3	Lb	Length	30	um
4	wg_width	Length	0.45	um
5	wg_height	Length	0.22	um
6	wg_angle	Number	90	
7	gap	Length	0.15	um
8	sep	Length	4	um
9	wavelength	Length	1.55	um
10	n_cells	Number	20	
11	n_cells_side	Number	20	

7. Run the simulation

   Running the EME simulation computes the waveguide modes and the coupling between them. We will be using the EME Analysis Window for our device design.



8. Perform a propagation sweep

Check *Propagation Sweep* in the EME analysis window. Set the parameter to ***group span 2***. Set the start to **0 µm** and stop to **50 µm**, with the interval at **1 µm**. Run the *eme sweep* and *visualize*.



Remove all the datasets except for **s31** (through port) and **s41** (drop port). Set the scalar operation to **Abs^2**. Here I changed the color of s41 to green for better visualization.

Questions:

- At what length(s) do we achieve 50:50 power splitting?
- At what length(s) does all the light couple into the drop port?
- At what length(s) does all the light remain in the through port?
- Why is there already some light in the drop port when the length (length of the parallel waveguides region) is 0?

  → Observe how light propagates in the bending region near the parallel waveguides. Does coupling occur before the waveguides become fully parallel?

9. Run a wavelength sweep

Select *wavelength sweep*. Set the start wavelength to **1.5 µm** and the stop wavelength to **1.6 µm**. Set the number of points to be 100. Run and visualize the sweep.



Remove all the datasets except for **s31** (through port) and **s41** (drop port). Set the scalar operation to **Abs^2**.

Questions:

- At what wavelength does the directional coupler work perfectly at a 50:50 power splitting ratio?
- If we can accept a ±10% variation in the power splitting ratio, what is the wavelength bandwidth of this directional coupler?

10. Click *eme propagate* in the *EME Analysis window*, then return to the model to visualize the *field profile* captured by the monitor.





Follow-up Questions

11. Examine the EME solver:
    - Vary the number of FDE cells for a convergence test. Which part of the model is more sensitive to the number of cells?
    - Set the number of cells for the parallel region to **1** and re-run the simulation. Observe the results. Why does the system still work with only one cell for the parallel region?

      → Do we need multiple cells for regions with a steady cross-section along the z-direction (propagation direction)?

12. Analyze the system. How can we effectively design a 50/50 directional coupler splitter?
    a. Waveguide gap
    b. Waveguide thickness
    c. Waveguide length
    d. Waveguide index
    e. Other waveguide geometry
    f. Background index

13. How is the fabrication tolerance of the design? How sensitive is it to fabrication variation?

14. What properties do we want for a directional coupler? How can we make it broadband or less sensitive to fabrication errors?

Broadband behavior analysis for current setup

From the propagation sweep, we observe that there are multiple coupling lengths where the power splitting ratio reaches approximately 50:50. Which of these — shorter or longer coupling lengths — provides a broader bandwidth?

For example, 50:50 power splitting occurs when the parallel region is around 0.1 µm, 16.8 µm, 33.5 µm, etc. The following wavelength sweep (from 1.5 µm to 1.6 µm) compares the power splitting ratio for directional couplers with parallel regions of 0.1 µm and 33.5 µm.

From the plot, which configuration exhibits less variation in splitting ratio across the wavelength range?





Why is the shorter coupling length generally more broadband? A detailed analysis of this result will be provided in Part 2.